text: # Maintainer: ArchStrike <team@archstrike.org>
  
  buildarch=220
  
  pkgname="ftpmap"
  pkgver=0.4
  pkgrel=5
  groups=('archstrike' 'archstrike-scanners' 'archstrike-fingerprinting')
  pkgdesc='scans remote FTP servers to identify what software and what versions
  they are running.'
  arch=('i686' 'x86_64' 'armv6h' 'armv7h' 'aarch64')
  url='http://wcoserver.googlecode.com/files/'
  license=('GPL2')
  source=("https://storage.googleapis.com/google-code-archive-downloads/v2/code.google.com/wcoserver/ftpmap-0.4.tar.gz"
          "config.guess-for-aarch64.patch"
  	"fix-implicit-time-include.patch")
  sha512sums=('eb2f6fa01071e6de7a87cb159e9f3c7b6f8258ac76177c7b339e63d2a89fc9884162ec0bff0495e7383deea76275370ff2fbf740c2960d28bba909a0c936bd10'
              '8fbc85fcd0efcc15b2217f752b6b0e4db8fef67ee5c444b74691e9ef619c951cbcf9d33e386a2995e9a0b6210cf3b8f65103f6011658441e52b9acab8e2546db'
              '015bb70729322bb2273dbe3bbffe575a62df9baeeaf6154af375669d753b3a8b9b24b49e91c3fcfc54220d81c4db5723d2df81f46ad163e0c9a20f642fc549b1')
  
  prepare() {
    cd "${srcdir}/${pkgname}-${pkgver}"
    patch -Np1 -i "${srcdir}/config.guess-for-aarch64.patch"
    cd src
    patch -i "${srcdir}/fix-implicit-time-include.patch"
  }
  
  
  build() {
    cd "${srcdir}/${pkgname}-${pkgver}"
  
    ./configure --prefix=/usr
    make
  }
  
  package() {
    cd "${srcdir}/${pkgname}-${pkgver}"
  
    install -Dm755 "src/${pkgname}" "${pkgdir}/usr/bin/${pkgname}"
    install -Dm644 "COPYING" "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
  }
  